\Orchestrating a brighter world NEC

Vector Engine向け Fortranコンパイラの使い方

第2版 2019年 11月発行

Orchestrating a brighter world

未来に向かい、人が生きる、豊かに生きるために欠かせないもの。
それは「安全」「安心」「効率」「公平」という価値が実現された社会です。

NECは、ネットワーク技術とコンピューティング技術をあわせ持つ
類のないインテグレーターとしてリーダーシップを発揮し、
卓越した技術とさまざまな知見やアイデアを融合することで、
世界の国々や地域の人々と協奏しながら、
明るく希望に満ちた暮らしと社会を実現し、未来につなげていきます。

目次

なお、本書で「並列処理」と記述したとき、コンパイラの自動並列化機能、または、OpenMP Fortran機能を使用した共有メモリ型並列処理を指します。

Vector Engine向けFortranコンパイラ

製品名：NEC Fortran Compiler for Vector Engine

- 対応する言語仕様
  - ISO/IEC 1539-1:2004 Programming languages – Fortran
  - ISO/IEC 1539-1:2010 Programming languages – Fortranの一部機能
  - OpenMP Version 4.5
- 主な機能
  - 自動ベクトル化機能
  - 自動並列化機能・OpenMP Fortran
  - 自動インライン展開機能

Fortranコンパイラの使い方

Fortranコンパイラの利用

```
$ nfort -mparallel -O3 a.f90 b.f90
```
… Fortranプログラム(a.f90, b.f90)のコンパイル、リンク

-O4 … 最大レベルの自動ベクトル化を適用
-O3 … 高度なレベルの自動ベクトル化を適用
-O2 … 既定レベルの自動ベクトル化を適用
-O1 … 副作用のない自動ベクトル化を適用
-O0 … ベクトル化、最適化を行わない

高 ↕ 低

これらは、コンパイラの自動ベクトル化、最適化レベルをコントロールする、

-fopenmp … OpenMP Fortran機能を利用
-mparallel … 自動並列化機能を利用

これらは、コンパイラの並列処理機能をコントロールする。
並列処理機能を使用しないときは指定しなくてよい。

代表的なコンパイラオプションの指定例

`$ nfort a.f90`	既定レベルの自動ベクトル化を適用し、コンパイル、リンク
`$ nfort -O4 a.f90 b.f90`	最大レベルの自動ベクトル化を適用し、複数のプログラムをコンパイル、リンク
`$ nfort -mparallel -O3 a.f90`	自動並列化、および、高度なレベルの自動ベクトル化を適用し、コンパイル、リンク
`$ nfort -O4 -finline-functions a.f90`	自動インライン展開、および、最大レベルの自動ベクトル化を適用し、コンパイル、リンク
`$ nfort -O0 -g a.f90`	ベクトル化を止めて、シンボリックデバッグするコンパイル、リンク
`$ nfort -g a.f90`	ベクトル化を止めずに、シンボリックデバッグするコンパイル、リンク
`$ nfort -E a.f90`	プリプロセスのみ実行。プリプロセス結果は標準出力に出力する
`$ nfort -fsyntax-only a.f90`	シンタックスチェックのみ実行

プログラムの実行

```
$ nfort a.f90 b.f90
$ ./a.out
```

実行ファイルを指定

```
$ ./b.out data1.in
```

実行ファイルに入力ファイルやパラメータを渡すとき、実行ファイル名に続けてそれらを指定

```
$ ./c.out < data2.in
```

実行ファイルに入力ファイルをリダイレクト

```
$ nfort -mparallel -O3 a.f90 b.f90
$ export OMP_NUM_THREADS=4
$ ./a.out
```

自動並列化したプログラムのスレッド数は環境変数**OMP_NUM_THREADS**で指定できる

```
$ env VE_NODE_NUMBER=1 ./a.out
```

利用するVEノードは
環境変数**VE_NODE_NUMBER**で指定する

実行性能の測定方法

Vector Engineでの性能測定機能

PROGINF(プログインフ)

- プログラム全体の性能情報
- 性能情報取得のためのオーバーヘッドは極小

FTRACE(エフトレース)

- 関数ごとの性能情報
- プログラムの再コンパイル、再リンクが必要
- 関数の呼び出し回数が多いと、性能情報取得のためのオーバーヘッドが大きくなり、実行時間が長くなることがある

PROGINF

プログラム全体の性能情報

```
$ nfort -O4 a.f90 b.f90 c.f90
$ ls a.out
a.out
$ export VE_PROGINF=DETAIL
$ ./a.out
            ********  Program  Information  ********
  Real Time (sec)                          :            11.329254
  User Time (sec)                          :            11.323691
  Vector Time (sec)                        :            11.012581
  Inst. Count                              :          6206113403
  V. Inst. Count                           :          2653887022
  V. Element Count                         :        619700067996
  V. Load Element Count                    :          53789940198
  FLOP count                               :        576929115066
  MOPS                                     :         73492.138481
  MOPS (Real)                              :         73417.293683
  MFLOPS                                   :         50976.512081
  MFLOPS (Real)                            :         50924.597321
  A. V. Length                             :           233.506575
  V. Op. Ratio (%)                         :            99.572922
  L1 Cache Miss (sec)                      :             0.010847
  CPU Port Conf. (sec)                     :             0.000000
  V. Arith. Exec. (sec)                    :             8.406444
  V. Load Exec. (sec)                      :             1.384491
  VLD LLC Hit Element Ratio (%)            :           100.000000
  Power Throttling (sec)                   :             0.000000
  Thermal Throttling (sec)                 :             0.000000
  Max Active Threads                       :                    1
  Available CPU Cores                      :                    8
  Average CPU Cores Used                   :             0.999509
  Memory Size Used (MB)                    :           204.000000
```

実行時に環境変数**VE_PROGINF**に
以下のどちらかの値をセット
“**YES**” … 基本情報
“**DETAIL**” … 基本情報+メモリ情報

時間情報

命令実行回数情報

ベクトル化情報・メモリ情報・
並列化情報

FTRACE

```
$ nfort -ftrace a.f90 b.f90 c.f90
```
(コンパイル、リンク時に**-ftrace**コンパイラオプションを指定)

```
$ ./a.out
$ ls ftrace.out
ftrace.out
```
(実行時、性能情報が格納されたftrace.outファイルを出力)

```
$ ftrace
```
(**ftrace**コマンドで解析結果を表示)

```
*----------------------*
  FTRACE ANALYSIS LIST
*----------------------*

Execution Date : Thu Mar 22 17:32:54 2018 JST
Total CPU Time : 0:00'11"163 (11.163 sec.)


FREQUENCY  EXCLUSIVE       AVER.TIME     MOPS   MFLOPS  V.OP  AVER.    VECTOR L1CACHE CPU PORT VLD LLC PROC.NAME
           TIME[sec](  % )    [msec]                    RATIO V.LEN      TIME    MISS     CONF HIT E.%

    15000     4.762( 42.7)     0.317  77117.2  62034.6  99.45 251.0     4.605   0.002    0.000  100.00 FUNC_A
    15000     3.541( 31.7)     0.236  73510.3  56944.5  99.46 216.0     3.554   0.000    0.000  100.00 FUNC_B
    15000     2.726( 24.4)     0.182  71930.2  27556.5  99.43 230.8     2.725   0.000    0.000  100.00 FUNC_C
        1     0.134(  1.2)   133.700  60368.8  35641.2  98.53 214.9     0.118   0.000    0.000    0.00 MAIN
-------------------------------------------------------------------------------------------------------------
    45001    11.163(100.0)     0.248  74505.7  51683.9  99.44 233.5    11.002   0.002    0.000  100.00 total
```

MPIプログラムのとき、性能情報が格納されたファイルが複数出力される。それらを**-f**オプションで指定する。

```
$ ls ftrace.out.*
ftrace.out.0.0   ftrace.out.0.1   ftrace.out.0.2   ftrace.out.0.3
$ ftrace -f ftrace.out.0.0 ftrace.out.0.1 ftrace.out.0.2 ftrace.out.0.3
```

性能測定時の注意

- FTRACEでは、手続の入口/出口で性能情報を採取するため、手続の呼び出し回数が多いプログラムでプログラム全体の実行時間が増加してしまう。

```
$ nfort -ftrace -c a.f90
$ nfort -c main.f90 b.f90 c.f90
$ nfort -ftrace a.o main.o b.o c.o
$ ./a.out
```

- 目的の関数が含まれているファイルのみ**-ftrace**付きでコンパイルする
- リンク時にも**-ftrace**を指定する

- **-ftrace**なしでコンパイルされたファイル中の手続の性能情報は、それらを呼び出している手続の性能情報に含めて表示される
- システムライブラリ関数に関する性能情報
  - PROGINFで表示される性能情報には、プログラムから呼び出しているシステムライブラリ関数の性能情報も含まれる
  - FTRACEで表示される性能情報には、プログラムから呼び出しているシステムライブラリ関数の性能情報も含まれる。それらは、呼び出した手続の性能情報に含めて表示される

プログラムのデバッグ

トレースバック機能

トレースバック機能を利用する場合、コンパイル、リンク時に**-traceback**を指定し、実行時に環境変数**VE_TRACEBACK**に“**FULL**”をセット

演算例外を発生させるには
環境変数**VE_FPE_ENABLE**に、
以下の値のいずれかをセット
“**DIV**” … ゼロ除算例外
“**INV**” … 無効演算例外

※ VE_FPE_ENABLEは上記以外の値も設定できるが、トレースバックでは基本的に上記二つを使用する

```
PROGRAM MAIN
REAL :: A, B
A = 1.0
B = 0.0
PRINT *, A/B
END
```

ゼロ除算が発生

```
$ nfort -traceback main.f90
$ export VE_TRACEBACK=FULL
$ export VE_ADVANCEOFF=YES
$ export VE_FPE_ENABLE=DIV
$ ./a.out
Runtime Error: Divide by zero at 0x6000000105d0
[ 1] Called from 0x600000010750
[ 2] Called from 0x7f8f41e307a8
[ 3] Called from 0x600000003700
Floating point exception
$ naddr2line -e a.out -a 0x6000000105d0
0x00006000000105d0
/.../main.f90:5
```

コンパイル、リンク時に**-traceback**を指定
トレースバック機能を使用
先行命令制御機構をOFF
ゼロ除算時に例外発生

トレースバック情報

ソースコード中の例外発生箇所を特定

main.f90ファイル内の5行目でゼロ除算が発生していると分かる

デバッガ(gdb)の利用

実行時間の長いプログラムでは、事前に問題のある関数を突き止めて置き、
その関数が含まれるファイルのコンパイル時のみ**-g**オプションを使用する

```
$ nfort -O0 -g -c a.f90
$ nfort -O4 -c b.f90 c.f90
$ nfort a.o b.o c.o
$ gdb a.out
(gdb) break sub
Breakpoint 1 at sub
(gdb) run
Breakpoint 1 at sub
(gdb) continue
...
```



注意事項

- **-O0**を指定せずにデバッグするとき、コンパイラの最適化によりコードや変数が削除、移動されるため、デバッガで変数が参照できなかったり、ブレークポイントが設定できないことがある。
- HWによる命令の先行制御によって例外発生個所が正しく表示されないことがある。環境変数**VE_ADVANCEOFF**に “**YES**”を設定することで先行制御を無効にできる。ただし、先行制御を無効にすることでプログラムの実行時間が大幅に長くなることがあるため注意すること。

システムコールのトレース:strace

```
$ /opt/nec/ve/bin/strace ./a.out
...
write(2, "delt=0.0251953, TSTEP".., 27)            = 27
open("MULNET.DAT", O_WRONLY|O_CREAT|O_TRUNC, 0666)= 5
ioctl(5, TCGETA, 0x8000000CC0)                     Err#25 ENOTTY
fxstat(5, 0x8000000AB0)                            = 0
write(5, "1 2 66 65", 4095)                        = 4095
write(5, "343 342", 4096)                          = 4096
write(5, "603 602", 4096)                          = 4096
write(5, "863 862", 4094)                          = 4094
write(5, "1105 1104", 4095)                        = 4095
write(5, "1249 1313 1312", 4095)                   = 4095
write(5, "1456 1457 1521 1520", 4095)             = 4095
write(5, "1727", 4095)                             = 4095
...
```

システムコールの引数　　システムコールの返却値

システムコールの引数、返却値のトレース情報の表示

- システムライブラリの呼び出しが適切に行われたか? などが確認できる。
- 出力が大量になるので、**strace**コマンドの**-e**オプションでトレースするシステムコールを厳選するとよい。

自動ベクトル化機能

ベクトル化とは?

規則的に並んだデータ列をベクトルデータと呼び、ベクトルデータを処理するスカラ命令列を、等価な処理を行うベクトル命令で置き換えることをベクトル化という

スカラ命令の実行イメージ

```
A(1) = B(1) + C(1)
A(2) = B(2) + C(2)
A(3) = B(3) + C(3)
…
A(100)=B(100) + C(100)
```



ベクトル命令の実行イメージ

```
DO I = 1, 100
  A(I) = B(I) + C(I)
END DO
```



HW命令との対応

スカラ機のときこの四つの命令列を
100回繰り返さなければならない

```
A(1)   =  B(1)    +  C(1)
A(2)   =  B(2)    +  C(2)
…
A(100)=   B(100)  +  C(100)
④          ①        ③     ②
```

```
① VLoad   $vr1, B(1:100)
② VLoad   $vr2, C(1:100)
③ VAdd    $vr3, $vr1, $vr2
④ VStore  $vr3, A(1:100)
```



VEでは、最大256個の配列要素をまとめてベクトルレジスタに取り込み、一度に計算を実行できる

命令実行時間の比較

```
DO I = 1, 100
  A(I) = B(I) + C(I)
  D(I) = E(I) + F(I)
END DO
```

スカラ加算命令の実行イメージ(2命令同時実行時)



ベクトル加算命令の実行イメージ

ベクトル命令はループの繰り返し数が十分大きいとき、その最大性能を発揮できる

ベクトル化できるループ

- ベクトル化に適合する演算、型のみを含むループ
  - 文字型、4倍精度実数型、2バイトの整数型を含まない
    - 数値計算では、ほとんど使われない型
    - 対応する型のベクトル演算命令がないためベクトル化できない
  - 手続呼び出しを含まない
    - 三角関数、指数関数、対数関数等を除く。これらは、ベクトル処理可能
- 配列や変数の定義・参照関係に、ベクトル化を阻害する依存関係(ベクトル化不可の依存関係)がない
  - 計算順序の変更が可能であること
- ベクトル化によって、性能の向上が期待できる
  - ループ長(ループの繰り返し数)が十分に大きい

ベクトル化不可の依存関係 (1)

以前の繰り返しで定義された配列要素や変数を、後の繰り返しで参照するパターンのとき、計算順序を変更できない

例1

```
DO I = 2, N
  A(I+1) = A(I) * B(I) + C(I)
END DO
```

ベクトル化すると、更新されたAの値が参照できないので、ベクトル化できない

スカラでの計算順序

```
a(3) = A(2) * B(2) + C(2)
a(4) = a(3) * B(3) + C(3)
a(5) = a(4) * B(4) + C(4)
a(6) = a(5) * B(5) + C(5)
  :
```

ベクトルでの計算順序

```
a(3) = A(2) * B(2) + C(2)
a(4) = A(3) * B(3) + C(3)
a(5) = A(4) * B(4) + C(4)
a(6) = A(5) * B(5) + C(5)
  :
```

a(n) : 更新されたAの値

更新前の値

例2

```
DO I = 2, N
  A(I-1) = A(I) * B(I) + C(I)
END DO
```

ベクトル化しても、計算順序は変わらないので、ベクトル化できる

スカラでの計算順序

```
a(1) = A(2) * B(2) + C(2)
a(2) = A(3) * B(3) + C(3)
a(3) = A(4) * B(4) + C(4)
a(4) = A(5) * B(5) + C(5)
  :
```

ベクトルでの計算順序

```
a(1) = A(2) * B(2) + C(2)
a(2) = A(3) * B(3) + C(3)
a(3) = A(4) * B(4) + C(4)
a(4) = A(5) * B(5) + C(5)
  :
```

ループの繰り返し間で、右下向きの矢印ができないかに注目する

ベクトル化不可の依存関係 (2)

例3

```
DO I = 1, N
   A(I) = S
   S = B(I) + C(I)
END DO
```

変数の参照が、定義よりも先に
現れるループはベクトル化できない

```
A(1) = S
DO I = 2, N
   S = B(I-1) + C(I-1)
   A(I) = S
END DO
S = B(N) + C(N)
```

プログラムを変形することで
ベクトル化できる

スカラでの計算順序

```
A(1) = S
S = B(1) + C(1)
A(1) = S
S = B(1) + C(1)
     :
```

ベクトルでの計算順序

```
A(1) = S
A(2) = S
     :
A(N) = S
S = B(1) + C(1)
S = B(2) + C(2)
     :
```

スカラでの実行順序

```
A(1) = S
S = B(1) + C(1)
A(2) = S
S = B(2) + C(2)
     :
```

ベクトルでの実行順序

```
A(1) = S
S = B(1) + C(1)
S = B(2) + C(2)
     :
A(2) = S
A(3) = S
     :
```

ベクトル化不可の依存関係 (3)

例4

```
S = 1.0
DO I = 1, N
   IF (A(I) .LT. 0.0) THEN
      S = A(I)
   END IF
   B(I) = S + C(I)
END DO
```

変数の参照の前に定義があっても、定義が実行されない可能性があるとベクトル化できない

例5

```
DO I = 1, N
   IF (A(I) .LT. 0.0) THEN
      S = A(I)
   ELSE
      S = D(I)
   END IF
   B(I) = S + C(I)
END DO
```

Sの参照の前に必ずSの定義があるのでベクトル化できる

例6

```
DO I = 1, N
   A(I) = A(I+K) + B(I)
END DO
```

コンパイル時にKの値が不明なため依存関係の有無が判定できないので、ベクトル化できない

(例1のパターンか例2のパターンか不明)

配列式のベクトル化

ソースプログラム

```
A(1:M,1:N) = B(1:M,1:N) + C(1:M,1:N)
B(1:M,1:N) = SIN(D(1:M,1:N))
```

コンパイラによる変形イメージ１

```
DO J = 1, N
  DO I =1, M
    A(I,J) = B(I,J) + C(I,J)
  END DO
END DO
DO J = 1, N
  DO I =1, M
    B(I,J) = SIN(D(I,J))
  END DO
END DO
```

コンパイラによる変形イメージ２

```
DO J = 1, N
  DO I =1, M
    A(I,J) = B(I,J) + C(I,J)
    B(I,J) = SIN(D(I,J))
  END DO
END DO
```

配列式は、内部的には**DO**ループと同様に変形してから、ループ融合、一重化などの最適化を行い、最適な次元でベクトル化される。

IF文のベクトル化

条件分岐(IF文)もベクトル化される。

```
DO I = 1, 100
  IF (A(I) .LT. B(I)) THEN
    A(I) = B(I) + C(I)
  END IF
END DO
```



ベクトル実行

```
mask(1)   = A(1) .LT. B(1)
mask(2)   = A(2) .LT. B(2)
  :          :          :
mask(100) = A(100) .LT. B(100)
```

```
if (mask(1) == .TRUE.)   A(1) = B(1) + C(1)
if (mask(2) == .TRUE.)   A(2) = B(2) + C(2)
 :             :             :
if (mask(100) == .TRUE.) A(100) = B(100) + C(100)
```

ベクトル化診断メッセージ

- コンパイラの出力するメッセージ、リストにより、ループのベクトル化状況、ベクトル化不可原因を調べることができる
  - 標準エラー出力　　… **-fdiag-vector=2** (詳細情報出力)
  - リストファイル出力　… **-report-diagnostics**

ベクトル化不可と思われる依存関係が変数RHOにあったとみなし、ベクトル化しなかったことを示すメッセージ

```
$ nfort –fdiag-vector=2 abc.f
…
nfort: vec( 103): abc.f, line 23: Unvectorized loop.
nfort: vec( 122): abc.f, line 24: Dependency unknown. Unvectorizable dependency is assumed.: RHO
nfort: vec( 122): abc.f, line 25: Dependency unknown. Unvectorizable dependency is assumed.: RHO
nfort: vec( 101): abc.f, line 50: Vectorized loop.
…
$ nfort –report-diagnostics abc.f
…
$ less abc.L
FILE NAME: abc.f
…
PROCEDURE NAME: SUB
DIAGNOSTIC LIST

 LINE              DIAGNOSTIC MESSAGE

    23: vec( 103): Unvectorized loop.
    24: vec( 122): Dependency unknown. Unvectorizable dependency is assumed.: RHO
    25: vec( 122): Dependency unknown. Unvectorizable dependency is assumed.: RHO
    50: vec( 101): Vectorized loop.
…
```

リストファイル名は「ソースファイル名.L」

編集リスト

ソース行とともにループ構造、そのベクトル化状況などを記号で表示

- **-report-format**が指定されたとき出力

```
$ nfort -report-format a.f90 -c
...
$ less a.L
          :
PROCEDURE NAME: SUB
FORMAT LIST

 LINE    LOOP       STATEMENT

     1:             SUBROUTINE SUB(A, B, C, X, Y, Z, N)
     2:               INTEGER :: N
     3:               REAL(KIND=4) :: A(N), B(N), C(N)
     4:               REAL(KIND=16) :: X(N), Y(N), Z(N)
     5:               INTEGER :: I
     6:
     7: V------>      DO I = 1, N
     8: |               A(I) = B(I) * C(I)
     9: V------       END DO
    10:
    11: +------>      DO I = 1, N
    12: |               X(I) = Y(I) * Z(I)
    13: +------       END DO
    14:
    15:             END SUBROUTINE SUB
```

リストファイル名は「ソースファイル名.L」

ベクトル化されたループ

ベクトル化されなかったループ

拡張ベクトル化機能

拡張ベクトル化機能とは

- そのままではベクトル化できない場合や、より効率のよいベクトル化が可能な場合に、コンパイラがプログラムを内部的に変形することで、ベクトル化の効果をさらに高める機能

- 文の入れ換え
- 多重ループの一重化
- 多重ループの入れ換え
- 部分ベクトル化
- 条件ベクトル化
- マクロ演算の認識
- 多重ループのベクトル化
- ループ融合
- インライン展開

文の入れ換え

ソースプログラム

```
DO I = 1, 99
   A(I) = 2.0
   B(I) = A(I+1)
END DO
```

コンパイラによる変形イメージ

```
DO I = 1, 99
   B(I) = A(I+1)
   A(I) = 2.0
END DO
```

そのままベクトル化すると、B(1)～B(99)の値がすべて2.0になってしまうので、このままではベクトル化不可。

ループ内の文の順序を入れ換えることにより、ベクトル化できるように変形。

多重ループの一重化

ソースプログラム

```
REAL A(M,N), B(M,N), C(M,N)
DO J = 1, N
  DO I = 1, M
    A(I,J) = B(I,J) + C(I,J)
  END DO
END DO
```

コンパイラによる変形イメージ

```
REAL A(M,N), B(M,N), C(M,N)
DO IJ = 1, M*N
  A(IJ,1) = B(IJ,1) + C(IJ,1)
END DO
```

ループ長（ループの繰り返し数）がより長くなるように、多重ループを一重化し、ベクトル命令の効率を高める。

多重ループの入れ換え

ソースプログラム

```
DO J = 1, M
  DO I = 1, N
    A(I+1,J) = A(I,J) + B(I,J)
  END DO
END DO
```

コンパイラによる変形イメージ

```
DO I = 1, N
  DO J = 1, M
    A(I+1,J) = A(I,J) + B(I,J)
  END DO
END DO
```



DO I=1,N でベクトル化しようとすると、配列Aにベクトル化不可の依存関係があり、ベクトル化できない。

```
A(2,1) = A(1,1) + B(1,1)
A(2,2) = A(1,2) + B(1,2)
A(2,3) = A(1,3) + B(1,3)
A(2,4) = A(1,4) + B(1,4)
```

ループを入れ換えると、 DO J=1,M のループに関してはベクトル化不可の依存関係がなくなり、ベクトル化できる。

部分ベクトル化

ソースプログラム

```
DO I = 1, N
  X = A(I) + B(I)
  Y = C(I) + D(I)
  WRITE(6,*) X, Y
END DO
```

コンパイラによる変形イメージ

```
DO I = 1, N
  WX(I) = A(I) + B(I)
  WY(I) = C(I) + D(I)
END DO
DO I = 1, N
  WRITE(6,*) WX(I), WY(I)
END DO
```

ベクトル化可能

ベクトル化不可

ループ構造に、ベクトル化できる部分とベクトル化できない部分が含まれている場合、ベクトル化可能な部分と不可能な部分に分割し、可能な部分だけをベクトル化する。

このとき必要であれば、作業ベクトル(上の例では配列WX、WY)を使用する。

条件ベクトル化

ソースプログラム

```
DO I = N, N+99
  A(I) = A(I+K) + B(I)
END DO
```

コンパイラによる変形イメージ

```
IF((K.GE.0) .OR. (K.LT.-99)) THEN
  ! ベクトル化したコード
ELSE
  ! ベクトル化しないコード
END IF
```

一つのループに対してベクトル化したコードとスカラのコード、特定のパターンのみ高速に実行できるコードなど、数種類のコードを用意し、実行時に条件を調べて、適切なコードを選択して実行するようにループを変形する。

(K=-1のとき)

```
A(I) = A(I-1) + B(I)
```

(A=-100のとき)

```
A(I) = A(I-100) + B(I)
```



マクロ演算の認識

総和型

```
DO I = 1, N
  S = S + A(I)
END DO
```

漸化式型

```
DO I = 1, N
  A(I) = A(I-1) * B(I) + C(I)
END DO
```

最大/最小型

```
DO I = 1, N
  IF (XMAX .LT. X(I)) THEN
    XMAX = X(I)
  END IF
END DO
```

配列や変数の定義・参照関係にベクトル化を阻害する依存関係があり、本来はベクトル化できない場合でも、コンパイラが特別なパターンであることを認識し、特別なベクトル命令を用いることで、ベクトル化を行う。

外側ループのベクトル化

ソースプログラム

```
DO I = 1, N
  DO J = 1, N
    A(I,J) = 0.0
  END DO
  B(I) = 1.0
END DO
```

コンパイラによる変形イメージ

```
DO I = 1,N
  DO J = 1,N
    A(I,J) = 0.0
  END DO
END DO
DO I = 1, N
  B(I) = 1.0
END DO
```

この例ではさらにループの一重化が適用される

ベクトル化は、基本的に最内側ループをベクトル化するが、外側のループを二つに分割することによって、外側のループにのみ含まれる文もベクトル化を行う。

ループ融合

ソースプログラム

```
DO I = 1, N
  A(I) = B(I) + C(I)
END DO
DO I = 1, N
  D(I) = SIN(E(I))
END DO
```

コンパイラによる変形イメージ

```
DO I = 1, N
  A(I) = B(I) + C(I)
  D(I) = SIN(E(I))
END DO
```

```
A(1:M) = B(1:M) + C(1:M)
D(1:M) = E(1:M) * F(1:M) + S
```

```
DO I = 1, M
  A(I,J) = B(I,J) + C(I,J)
  D(I,J) = E(I,J) * F(I,J) + S
END DO
```

コンパイラは同じ繰り返し回数を持つ複数のループ同士、または同じ形状（次元数と各次元のサイズ）をもつ複数の配列式同士を一つにまとめてベクトル化する。
同じ形状の配列式、ループ構造が連続していれば融合するが、間に形状の異なる配列式やループ構造、他の文があると融合できない。
高速化のためには、なるべく同じ形状の配列式、ループ構造を連続させるほうがよい。

インライン展開によるベクトル化

ソースプログラム

```
DO I = 1, N
  CALL SUB(B(I),C(I))
  A(I) = B(I)
END DO
   :
SUBROUTINE SUB(X,Y)
  X = SIN(Y)
END
```

コンパイラによる変形イメージ

```
DO I=1,N
  B(I) = SIN(C(I))
  A(I) = B(I)
END DO
   :
SUBROUTINE SUB(X,Y)
  X = SIN(Y)
END
```

-finline-functionsを指定すると、可能であれば関数を呼び出し元にインライン展開する。
ループ中に関数の呼び出しがあれば、インライン展開後にベクトル化を試みる。

プログラムのチューニング

コンパイラオプションを追加指定したり、プログラムへのコンパイラ指示行の挿入などにより、
プログラムを高速化する(実行時間を短くする）ことを「チューニング」と呼びます。
チューニングにより、*Vector Engine*のHW性能を最大限まで引き出すことができます。

チューニングの観点

ベクトル化率を高める

- ベクトル化率とは、プログラム全体のうち、ベクトル命令で実行可能な部分の比率
- ベクトル化不可の要因を取り除き、ベクトル化を促進
  - ベクトル命令で実行可能な部分を増やす

ベクトル命令の効率を高める

- 一つのベクトル命令で処理されるデータの個数を増やす
  - ループの繰り返し数(ループ長)を大きくする
- ループの繰り返し数が極端に短いループはベクトル化をやめる
  - p.21「命令実行時間の比較」のシートを参照

メモリアクセスの効率を高める

- リストベクトルの使用を避ける

ベクトル化率

- プログラム全体のうち、ベクトル命令で実行可能な部分の比率



- 一般に、ベクトル化率を正確に求めることは困難であるため、ベクトル演算率で代用

$$\text{ベクトル演算率} = 100 \times \frac{\text{ベクトル命令で処理されたデータの個数}}{\text{全命令実行数} - \text{ベクトル命令実行数} + \text{ベクトル命令で処理されたデータの個数}}$$

ループの繰り返し数と実行時間

- ループの繰り返し数をできるだけ大きくした方が、ベクトル化による高速化の効果が大きい
  - 一つのベクトル命令で処理できるデータの個数が多くなる



繰り返し数をループごとに分析するのは困難

平均ベクトル長で分析

一つのベクトル命令が処理したデータの個数の平均。
最大256個である。

チューニングの手順

- 性能測定機能のプログラムの性能情報より、実行時間の長い関数、ベクトル演算率が低い、平均ベクトル長が短い関数を特定
  - PROGINF
    - プログラム全体の実行時間、ベクトル演算率、平均ベクトル長
  - FTRACE
    - 手続ごとの実行時間、実行回数、ベクトル演算率、平均ベクトル長

- 特定した手続のベクトル化診断メッセージを参照し、ベクトル化されていないループを特定

- コンパイラオプション、コンパイラ指示行等を挿入し、ベクトル化を促進

PROGINF情報の利用

出力例

```
   ********  Program  Information  ********
Real Time (sec)                 :           11.336602
User Time (sec)                 :           11.330778
Vector Time (sec)               :           11.018179
Inst. Count                     :         6206113403
V. Inst. Count                  :         2653887022
V. Element Count                :       619700067996
V. Load Element Count           :         53789940198
FLOP count                      :       576929115066
MOPS                            :        73455.206067
MOPS (Real)                     :        73370.001718
MFLOPS                          :        50950.894570
MFLOPS (Real)                   :        50891.794092
A. V. Length                    :          233.506575
V. Op. Ratio (%)                :           99.572922
L1 Cache Miss (sec)             :            0.010855
CPU Port Conf. (sec)            :            0.000000
V. Arith. Exec. (sec)           :            8.410951
V. Load Exec. (sec)             :            1.386046
VLD LLC Hit Element Ratio (%)   :          100.000000
Power Throttling (sec)          :            0.000000
Thermal Throttling (sec)        :            0.000000
Max Active Threads              :                   1
Available CPU Cores             :                   8
Average CPU Cores Used          :            0.999486
Memory Size Used (MB)           :          204.000000
```

A.V.Length (平均ベクトル長)

- ベクトル命令の効率を表す指標
- 大きいほどよい (最大256)
- 小さいとき、ベクトル化されたループの繰り返し数が少ない。大きくできないか検討する

V.Op.Ratio (ベクトル演算率)

- ベクトル命令で処理されたデータの比率
- 大きいほどよい (最大100)
- 小さいとき、ベクトル化されたループが少ない、あるいは、ループ自体がプログラム中に少ない。ベクトル化できるループが他にないか検討する

FTRACE

関数ごとに性能情報を採取する機能

- PROGINFと同じように、V.OP.RATIO (ベクトル演算率)、AVER.V. LEN (平均ベクトル長) に注目し、手続ごとに分析する

```
*----------------------*
  FTRACE ANALYSIS LIST
*----------------------*

Execution Date : Thu Mar 22 15:47:42 2018 JST
Total CPU Time : 0:00'11"168 (11.168 sec.)


FREQUENCY  EXCLUSIVE       AVER.TIME     MOPS   MFLOPS  V.OP  AVER.   VECTOR L1CACHE CPU PORT VLD LLC PROC.NAME
           TIME[sec](  % )    [msec]                    RATIO V.LEN     TIME    MISS     CONF HIT E.%

    15000     4.767( 42.7)     0.318  77030.2  61964.6  99.45 251.0    4.610   0.002    0.000  100.00 FUNC_A
    15000     3.541( 31.7)     0.236  73505.6  56940.8  99.46 216.0    3.555   0.000    0.000  100.00 FUNC_B
    15000     2.726( 24.4)     0.182  71930.1  27556.5  99.43 230.8    2.725   0.000    0.000  100.00 FUNC_C
        1     0.134(  1.2)   133.700  60368.9  35641.3  98.53 214.9    0.118   0.000    0.000    0.00 MAIN
-------------------------------------------------------------------------------------------------------------
    45001    11.168(100.0)     0.248  74468.3  51657.9  99.44 233.5   11.008   0.002    0.000  100.00 total
```

プログラムの
チューニング・テクニック

コンパイラ指示行

- 自動ベクトル化、最適化の効果を促進させるために、**!NEC$** で書くことで、コンパイル時にわからない情報を与える。これをコンパイラ指示行と呼ぶ
  - コンパイラ指示行の形式

    !NEC$△指示オプション (△:空白) ... [自由形式、固定形式]

    ***NEC$**△指示オプション (△:空白) ... [固定形式]

    cNEC$△指示オプション (△:空白) ... [固定形式]
  - 主なベクトル化指示オプション
    - **VECTOR/NOVECTOR** : 自動ベクトル化の対象とする/しない
    - **IVDEP** : ベクトル化不可の依存関係がない

```
!NEC$ IVDEP
DO I = 1, N
  A(IX(I)) = A(IX(I)) + B(I)
END DO
```

- ベクトル化指示オプションはループの直前に指定
- 空白で区切って指定する
- 指示行の直後のループにのみ効果がある

ベクトル化不可の依存関係の対処 (1)

ベクトル化率を高める

```
nfort: vec( 103): a.f, line 16: Unvectorized loop.
nfort: vec( 113): a.f, line 16: Overhead of loop division is too large.
nfort: vec( 121): a.f, line 18: Unvectorizable dependency.
```

部分ベクトル化を試みるため、このようなメッセージが表示されることがある(以降、省略)

変数Tが定義されるかどうかわからないのでベクトル化できない

Unvectorized Loop

```
DO I = 1, N
  IF (X(I).LT.S) THEN
    T = X(I)
  ELSE IF (X(I).GE.S) THEN
    T = -X(I)
  END IF
  Y(I) = T
END DO
```

Vectorized Loop

```
DO I = 1, N
  IF (X(I).LT.S) THEN
    T = X(I)
  ELSE
    T = -X(I)
  END IF
Y(I) = T
```

変数Tが必ず定義されるよう修正

総和型のマクロ演算と認識できない

Unvectorized Loop

```
DO I = 1, N
  IF (A(I).GT.0.0) THEN
    S = S + B(I)
  ELSE
    S = S + C(I)
  END IF
END DO
```

Vectorized Loop

```
DO I = 1, N
  IF (A(I).GT.0.0) THEN
    T = B(I)
  ELSE
    T = C(I)
  END IF
  S = S + T
END DO
```

総和型のマクロ演算とし、ベクトル化

〈ベクトル化後の診断メッセージ〉

```
nfort: vec( 101): a.f, line 16: Vectorized loop.
nfort: vec( 126): a.f, line 22: Idiom detected.: Sum.
```

総和計算は特別な*HW*命令を使用してベクトル化

ベクトル化不可の依存関係の対処 (2)

ベクトル化率を高める

```
nfort: vec( 103): dep.f90, line 5: Unvectorized loop.
nfort: vec( 122): dep.f90, line 6: Dependency unknown. Unvectorizable dependency is
assumed.: A
```

- ベクトル化不可の依存関係が仮定されたが、実際にはベクトル化不可の依存関係がないことがわかっているとき、**IVDEP**を指定する

Unvectorized Loop

```
SUBROUTINE SUB(A, B, C, N, K)
  INTEGER I, N, K
  REAL A(N), B(N), C(N)

  DO I = 1, N
    A(I+K) = A(I) + B(I)
  END DO
END SUBROUTINE SUB
```

A(I-1)=A(I)のパターンか、A(I+1)=A(I)のパターンか不明なのでベクトル化しない

Vectorized Loop

```
SUBROUTINE SUB(A, B, C, N, K)
  INTEGER I, N, K
  REAL A(N), B(N), C(N)
!NEC$ IVDEP
  DO I = 1, N
    A(I+K) = A(I) + B(I)
  END DO
END SUBROUTINE SUB
```

A(I-1)=A(I)のパターンであることが明らかなとき、**IVDEP**を指定してベクトル化

〈ベクトル化後の診断メッセージ〉

```
nfort: vec( 101): dep.f90, line 5: Vectorized loop.
```

インライン展開によるベクトル化促進

ベクトル化率を高める

```
nfort: vec( 110): a.f90, line 4: Vectorization obstructive procedure reference.: FUN
nfort: vec( 103): a.f90, line 4: Unvectorized loop.
nfort: opt(1025): a.f90, line 5: Reference to this procedure inhibits optimization.: FUN
```

- 関数呼び出しがベクトル化を妨げているとき出力される
- **-finline-functions**コンパイラオプションで関数のインライン展開を試みる

```
SUBROUTINE SUB(A, B, C, D, N)
  INTEGER I, N
  REAL A(N), B(N), C(N), D(N)
  DO I=1, N                    ! Unvectorized
    A(I) = FUN(B(I), C(I)) / D(I)
  END DO
END

FUNCTION FUN(X, Y)
  REAL X, Y
  FUN = SQRT(X) * Y
END FUNCTION FUN
```

SQRTは、ベクトル処理可能な関数でありベクトル化を妨げない

<コンパイラオプションを指定する>

```
$ nfort -finline-functions a.f90
```

<コンパイラの変形イメージ>

```
DO I=1, N                    ! Vectorized
  A(I)= SQRT(B(I))*C(I) / D(I)
END DO
```

```
nfort: vec( 101): func.f90, line 4: Vectorized loop.
nfort: inl(1222): func.f90, line 5: Inlined: FUN
```

添字式に配列の現れるループ

ベクトル化命令の効率を高める

```
nfort: vec( 101): a.f90, line 5: Vectorized loop.
nfort: vec( 126): a.f90, line 6: Idiom detected.: LIST VECTOR
```

リストベクトルを**IVDEP**を指定することでさらに高速化

- 添字式に配列が現れる配列をリストベクトルと呼ぶ。
- 両辺に同じリストベクトルが現れたとき、その依存関係が不明であるのでベクトル化できない。

Vectorized Loop（**LIST_VECTOR**指定行）

```
!NEC$ LIST_VECTOR
  DO I = 1, N
    A(IX(I)) = A(IX(I)) + B(I)
  END DO
```

Vectorized Loop（**IVDEP**指示行）

```
!NEC$ IVDEP
  DO I = 1, N
    A(IX(I)) = A(IX(I)) + B(I)
  END DO
```

LIST_VECTORを指定するとベクトル化できるが、配列Aの要素がループ中で2回以上定義されないとき、つまり、IX(I)の値が同じになるIがなければ、***IVDEP***を指定することで、より効率のよいベクトル命令列で計算できる

<**IVDEP**によるベクトル化後のメッセージ>

```
nfort: vec( 101): a.f90, line 5: Vectorized loop.
```

外側ループのアンローリング

ベクトル化命令の効率を高める

外側ループのアンロールすることでロード、ストア回数を減らす

- ループを展開することをアンローリングという
- 2つ以上のループのネストがあるとき外側ループをアンロールすることで内側ループのインダクション変数のみを使用するロード、ストア数を減らせる

```
DO J = 1, N
  DO I = 1, M
    A(I,J) = B(I,J) + C(I)
  END DO
END DO
```

OUTERLOOP_UNROLL(4)指示行挿入

括弧内にアンロール段数2^xを指定する

```
!NEC$ OUTERLOOP_UNROLL(4)
DO J = 1, N
  DO I = 1, M
    A(I,J) = B(I,J) + C(I)
  END DO
END DO
```

外側ループを4段にアンロール後のプログラム

```
DO J = 1, N%3
  DO I = 1, M
    A(I,J) = B(I,J) + C(I)
  END DO
END DO

DO J = N%3+1, N, 4
  DO I = 1, M
    A(I,J)   = B(I,J)   + C(I)
    A(I,J+1) = B(I,J+1) + C(I)
    A(I,J+2) = B(I,J+2) + C(I)
    A(I,J+3) = B(I,J+3) + C(I)
  END DO
END DO
```

配列Cのベクトルロード1回に付き4回ベクトル演算を行える

OUTERLOOP_UNROLL指示行、または、**-fouterloop-unroll**オプションを指定すると外側ループのループ長が短くなり、配列Cのベクトルロード回数が減る

<**OUTERLOOP_UNROLL**指示行による外側ループアンロール後のメッセージ>

```
nfort: opt(1592): a.f90, line 5: Outer loop unrolled inside inner loop.: J
nfort: vec( 101): a.f90, line 6: Vectorized loop.
```

繰り返し数の小さいループ

ベクトル化命令の効率を高める

繰り返し数が小さいとき、ループ制御の処理を省いて高速化

- 繰り返し数<=256　　… ショートループにし、ループ終了処理を削除
- 繰り返し数<<256　　… ループ展開し、計算部分以外を削除



自動ベクトル化における注意事項

自動ベクトル化のレベルと適用される最適化

- **-04**、**-03**、**-02**で自動ベクトル化のレベルを変更したとき、次のベクトル化/最適化が自動的に適用される

高 ⟷ 低

適用されるベクトル化・最適化	-04	-03	-02
条件ベクトル化によるベクトル化 (**-m[no-]vector-dependency-test**)	○	○	○
一重化、入れ換え、行列積ライブラリ関数置換によるベクトル化 (**-f[no-]loop-collapse**、**-f[no-]loop-interchange**、**-f[no-]matrix-matiply**)	○	○	–
ループ中の配列を指すポインタ間にはベクトル化不可の依存関係がないと仮定 (**-fnamed-[no]alias**)	○	○	–
外側ループのアンローリングを許可する (**-f[no]-outerloop-unroll**)	○	○	–

＊主なもののみ記載、()内は個別に指定するときのコンパイラオプション

ベクトル化による演算結果への影響

ベクトル化した場合としない場合で、演算結果が誤差範囲で異なることがある

- 最適化・ベクトル化による演算順序の変更や除算の乗算化により、情報落ち・桁落ち・丸め誤差が変わるため。
- ベクトル化された数学関数では、高速にベクトル計算できるようスカラ版の数学関数と異なる計算アルゴリズムを使用しているため。
- 整数型漸化式マクロ演算では、浮動小数点数のベクトル命令を使用するため、52ビットで表現できる整数値のみ計算可能。
- ベクトル融合積和演算(FMA)が使用された場合、途中の積算結果を丸めずに和算が行われるため、使用しない場合に対して異なる演算結果となる可能性がある

誤差が気になる場合

- **NOVECTOR**指示行でループがベクトル化されないようにする。
- **NOFMA**指示行でベクトル積和演算が行われないようにする。

```
!NEC$ NOVECTOR
DO I = 1, N
  SUM = SUM + A(I)
END DO
```

ベクトル化による実行時バスエラーの発生

4バイトアラインされた配列を8バイト要素のベクトル命令でロード/ストアしている可能性がある

- 以下の例では、sub.f90のコンパイル時に**-fdefault-real=8**が指定されているため、**REAL**型の配列**A**、**B**が8バイト要素のベクトル命令でロード/ストアされる。
- 8バイトのベクトルロード/ストア命令は8バイトアラインを要求するため、ロード/ストアする配列が4バイトアラインされている場合、実行時にメモリの不正アクセスによってバスエラーが発生する。

main.f90

```
PROGRAM MAIN
  REAL :: A(512), B(512)
  ...
  CALL SUB(A,B,512)
END
```

sub.f90

```
SUBROUTINE SUB(A, B, N)
  INTEGER :: N
  REAL :: A(N), B(N)
  B = A                    !!!<---vectorized
END SUBROUTINE SUB
```

```
$ nfort -c main.f90
$ nfort -c -fdefault-real=8 sub.f90
$ nfort main.o sub.o
$ ./a.out
Bus error
```

配列を明示的に4バイトデータ型として宣言するか、**NOVECTOR**指示行でベクトル化を抑止する

```
SUBROUTINE SUB(A, B, N)
  INTEGER :: N
  REAL(KIND=4) :: A(N), B(N)
  B = A
END SUBROUTINE SUB
```

明示的に4バイトデータ型を指定

```
SUBROUTINE SUB(A, B, N)
  INTEGER :: N
  REAL :: A(N), B(N)
!NEC$ NOVECTOR
  B = A
END SUBROUTINE SUB
```

NOVECTOR指示行を指定

自動並列化機能・OpenMP Fortran

並列処理とは

- 一つの仕事を分割し、複数のスレッドで同時に実行すること
  - ループの繰り返しを分割
  - プログラム内の一連の処理(文の集まり)を分割



シリアル実行

ループの繰り返しを四つに分割したときの例



並列実行

並列処理による実行時間の短縮

並列処理により経過時間が短縮される

- 総CPU時間は並列処理のためのオーバヘッドなどで増加する



プログラムの並列化

複数のスレッドで並列実行できるようプログラミングすること

- ループや文の集まりを抽出し、並列処理できるようにプログラムを変形
- 自動並列化やOpenMPで並列実行する実行コードを生成

例*1.* 自動並列化による並列化

```
SUBROUTINE SUB(A, N)
  INTEGER :: N, I, J
  REAL(KIND=8) :: A(N), B(N)
  REAL(KIND=8) :: SUM = 1.0

  DO J = 1, N
    DO I = 1, N
      SUM = SUM + A[j] + B[I]
    ENDDO
  ENDDO

  RETURN
END SUBROUTINE SUB
```

並列実行できるループを探す

-mparallelを指定して自動並列化を有効にする

```
$ nfort -mparallel a.f90
nfort: par(1801): a.f90, line 6: Parallel routine generated.: SUB$1
nfort: par(1803): a.f90, line 6: Parallelized by "do".
nfort: vec( 101): a.f90, line 7: Vectorized loop.
```

内側ループをベクトル化

DO文を並列化

ループを並列実行するために別関数にして切り出す

※ループ以外の部分は並列実行できないものとする

Vector Engineで利用できる並列化プログラミング

OpenMP Fortran

- プログラマが、並列実行できるループや文の集まりを抽出し、それらの並列化方法を示す指示行(OpenMPディレクティブ)を指定
- コンパイラが、その指示を元にプログラムを変形、並列処理制御のための指示行を挿入

自動並列化

- コンパイラが、並列実行できるループや文の集まりを抽出、プログラムを並列処理制御するように変形
- 前ページの「例1」のループの検出、プログラム変形、指示行の挿入のすべての作業をコンパイラが自動的に行う

プログラミング手法	ループ・文の集まりの抽出	指示行の挿入	プログラムの変形	難易度
OpenMP Fortran (`-fopenmp`)	○	○	–	高
自動並列化 (`-mparallel`)	–	–	–	低

○：プログラマによる作業が必須
–：コンパイラが自動的に実施するので不要

※ チューニング時には「–」の項であってもプログラマによる作業が必要になることがある

OpenMP並列化

OpenMP Fortran

```
$ nfort -fopenmp a.f90 b.f90
```

リンクのときにも-fopnmpを指定すること

共有メモリ型並列処理のための指示行・ライブラリなどの国際標準

- NEC Fortran Compiler for Vector Engineでは、OpenMP Version 4.5までの一部機能をサポート

プログラミング手法

- プログラマが、並列実行できるループや文の集まりを抽出し、それらの並列化方法を示す指示行(OpenMPディレクティブ)を指定
- コンパイラはその指示を元にプログラムを変形、並列処理制御のための処理を挿入
- **-fopenmp**を指定して、コンパイル、リンク

特徴

- プログラマが並列化部分を選択、指定できるため、自動並列化より高い性能向上が期待
- 並列化部分の切り出し、バリア同期、変数の共有属性にかかわるプログラム変形をコンパイラが行うため、プログラミングが容易

例:OpenMP Fortranによる記述

例1の関数SUBのOpenMP Fortranによる並列化

```
SUBROUTINE SUB(A, N)
  INTEGER :: N, I, J
  REAL(KIND=8) :: A(N), B(N)
  REAL(KIND=8) :: SUM = 1.0
!$OMP PARALLEL DO
  DO J = 1, N
    DO I = 1, N
      SUM = SUM + A(J) + B(I)
    ENDDO
  ENDDO

  RETURN
END SUBROUTINE SUB
```

OpenMP指示行を挿入

並列実行できるループを探す

-fopenmpを指定してOpenMP指示行を有効にする

```
$ nfort -fopenmp a.f90
nfort: par(1801): a.f90, line 5: Parallel routine generated.: SUB$1
nfort: par(1803): a.f90, line 6: Parallelized by "do".
nfort: vec( 101): a.f90, line 7: Vectorized loop.
```

コンパイラが並列実行できるようにプログラムを変形する

OpenMPの指示行は”!$OMP”に続けて並列化方法を指定する

```
!$OMP PARALLEL DO
```

PARALLEL

並列化区間の開始の指定

DO

DOループの並列化を指定

用語

- OpenMPスレッド (OpenMP thread)
  - 論理的な並列処理の単位。スレッドと略されることもある
- 並列リージョン (Parallel region)
  - 複数のOpenMPスレッドにより並列に実行される文の集まり
- 逐次リージョン (Serial region)
  - 並列リージョンの外側でマスタスレッドでのみ実行される文の集まり
- プライベート (Private)
  - 並列リージョンを実行するOpenMPスレッドのうちの一つのスレッドのみからアクセス可能であること
- 共有 (Shared)
  - 並列リージョンを実行するすべてのOpenMPスレッドからアクセス可能であること

OpenMPディレクティブ

よく使用されるもののみ説明

!$OMP PARALLEL DO [*schedule*句] [NOWAIT]

SCHEDULE(STATIC[,*size*]) … SCHEDULE(STATIC)が既定値

- *size*回の繰り返しをひとまとまりとし、OpenMPスレッドにラウンドロビン方式で割り当て実行する
- *size*の指定が省略されたとき、*size*をスレッド数で割った値が指定されたものとみなす

SCHEDULE(DYNAMIC[,*size*])

- *size*回の繰り返しをひとまとまりとし、OpenMPスレッドに動的に割り当て実行する
- *size*の指定が省略されたとき、1が指定されたものとみなす

SCHEDULE(RUNTIME)

- 環境変数**OMP_SCHEDULE**に設定されたスケジュール方法で実行する

NOWAIT

- 並列ループ終了時の暗黙のバリア同期を行わない

!$OMP SINGLE

ひとつのOpenMPスレッドでのみ実行する。マスタスレッドとは限らず、
一番最後にディレクティブに到達したタスクで実行する

!$OMP CRITICAL

同時に複数のOpenMPスレッドで実行しないようにする(排他制御)

自動並列化機能

自動並列化機能

自動並列化機能を利用したとき、「プログラムの並列化」で示したようにコンパイラがすべて自動で行う

```
$ nfort -mparallel a.f90 b.f90
```

リンクのときにも-mparallelを指定すること

-mparallelを指定してコンパイル、リンク

- 並列実行できるループや文の集まりを抽出し、並列処理できるようにプログラムを変形する
  - 並列化の阻害要因を含まないループの自動選択
  - 多重ループの外側ループを自動選択
    - –最内側ループはベクトル化を使って高速化

コンパイラ指示行による自動並列化の制御

- コンパイラ指示行の形式

  !NEC$△指示オプション　(△:空白)

- 主な指示オプション
  - **CONCURRENT/NOCONCURRENT** … 直後のループの並列化を許可する/しない
  - **CNCALL** … ループ中に手続の呼出しがあるときでも並列化を許可する

指示行による自動並列化の制御

NOCONCURRENT ... 直後のループの並列化を許可しない

```
CALL SUB(4)                    ! 手続呼出し
...
SUBROUTINE SUB(M)
  INTEGER :: M
    ...
!NEC$ NOCONCURRENT
    DO J = 1, M                ! 実はMの値が小さい
      DO I = 1, N
        A(I) = B(J) / C(J)
      ENDDO
    ENDDO
```

繰り返し数の少ないループが並列化されると、並列化のためのオーバヘッドの占める比率が大きく、性能が低下してしまうことがある

NOCONCURRENTで並列化を抑止

CNCALL ... ループ中に手続呼出しがあるときでも並列化を許可する

```
!NEC$ CNCALL
  DO I = 1, M
    A(I) = FUNC(B(I), C(I))
  ENDDO
```

手続が並列実行できるかどうかわからないため、手続呼び出しを含むループは自動並列化されない

手続が並列実行できるとき、**CNCALL**を指定

(手続が並列実行できることは、プログラマが保証しなければならない)

OpenMP・自動並列化機能の同時利用

```
$ nfort –fopenmp –mparallel a.f90 b.f90
```

-fopenmpと**-mparallel**の両方を指定してコンパイル、リンク

- OpenMP並列リージョン外のループが自動並列化の対処となる
- OpenMPディレクティブを含む手続を自動並列化したくないとき、**-mno-parallel-omp-routine**を指定する

```
SUBROUTINE SUB(A, N)
  INTEGER :: I, J, N
  REAL(KIND=8) :: A(N), B(N,N)
  REAL(KIND=8) :: S = 1.0

  DO I = 1, N
    DO J = 1, N
      B(J,I) = I * J
    END DO
  END DO

!$OMP PARALLEL DO
  DO J = 1, N
    DO I = 1, N
      S = S + A(J) + B(J,I)
    END DO
  END DO
...
END SUBROUTINE SUB
```

自動並列化される

OpenMP並列化される

```
$ nfort -fopenmp -mparallel t.f90
nfort: par(1801): t.f90, line 6: Parallel routine generated.: SUB$1
nfort: par(1803): t.f90, line 6: Parallelized by "do".
nfort: vec( 101): t.f90, line 7: Vectorized loop.
nfort: par(1801): t.f90, line 12: Parallel routine generated.: SUB$2
nfort: par(1803): t.f90, line 13: Parallelized by "do".
nfort: vec( 101): t.f90, line 14: Vectorized loop.
```

並列処理プログラムの動作

OpenMP並列化されたプログラムの実行イメージ

OpenMPを用いて並列化したとき

```
SUBROUTINE SUB (A, N)
  INTEGER :: N, I, J
  REAL(KIND=8) :: A(N), B(N)
  REAL(KIND=8) :: SUM = 1.0
  REAL(KIND=8) :: DERIVE
!$OMP PARALLEL PRIVATE(DERIVE)
  DERIVE = 12.3
!$OMP DO
  DO I = 1, N
    B(I) = DERIVE
  ENDDO
!$OMP END PARALLEL
  ...
!$OMP PARALLEL DO REDUCTION(+:SUM)
  DO J = 1, N
    DO I = 1, N
      SUM = SUM + A(J) + B(I)
    ENDDO
  END DO
  RETURN
END SUBROUTINE SUB
```



※ VEでは、ネスト並列 (nested parallelism) はサポートしていない。

自動並列化されたプログラムの実行イメージ

```
SUBROUTINE SUB (A, B, N)
  INTEGER :: N, I, J
  REAL(KIND=8) :: A(N), B(N)
  REAL(KIND=8) :: SUM = 0.0
  ...

 DO J = 1, N
   DO I = 1, N
     SUM = SUM + A(J) + B(I)
   ENDDO
 ENDDO

  RETURN

END SUBROUTINE SUB
```



(実線：プログラムの実行、破線：待ち合わせ処理)

OpenMP・自動並列でのスレッド数の決定

並列処理に使用するスレッド数は以下のルールで決定



※ VEのコア数は8個であるため、８より大きいスレッド数を指定しても
最大８スレッドまでしか生成されない。

並列処理プログラムのチューニング

チューニングの観点

並列実行される部分は多いか?

- 並列化しないで実行した場合の経過時間に対する、並列実行可能部分の実行時間の割合が小さくないか? (並列実行部分/並列ループを増やす)

効果的な並列化が行われているか?

- 並列化されているループの実行時間が長いか? (適切なループを並列化する)
- 並列化のためのオーバヘッドが大きくないか? (オーバヘッドを小さくする)
- スレッドごとの作業量が均一か? (ループ内の処理を見直す)



チューニング手順

使用する性能解析機能
- 簡易性能解析(**FTRACE**)

1. 並列対象ループ/関数の抽出
   - プロファイラ情報、FTRACEの出力から、実行時間の長い関数を見つけ出す
2. 並列化部分を増やす
   - **1.**で見つけた関数中の並列化されていないループが並列化できないか調べ、必要なら指示行の指定、プログラムの変形を行い並列化する
3. ロードバランスの改善
   - PROGINF情報、FTRACEの出力から、各タスクに処理が均等に割り当てられるよう、ロードバランスを調整する

※ 並列化の前にベクトル化のチューニングを十分行っておくこと

並列化するループの選択

自動並列化では該当するループが自動的に探し出され、並列化される

並列化阻害要因のないループ

- 並列化できない依存関係
- 並列化できない制御構造
- I/O関数など実行順序を保証しなければならない関数呼び出し

多重ループの最外側ループ

- 処理時間の長いループ
- 最内側ループはベクトル化での高速化を検討

```
DO J = 1, N
  DO I = 1, M
    A(I,J) = B(I,J) + C(I,J)
  ENDDO
ENDDO
```



並列化できない依存関係

同一配列要素が異なる繰り返しで定義・参照されているループ

同一配列要素の定義・参照

```
DO I = 1, N
   A(I) = B(I+1)
   B(I) = C(I)
ENDDO
```

ループの繰り返し	参照	定義	
I=1	B(2)	B(1)	スレッド1で実行
I=2	B(3)	B(2)	
I=3	B(4)	B(3)	スレッド2で実行
I=4	B(5)	B(4)	
⋮	⋮	⋮	

B(3)の参照と定義はどちらが先になるか保証されない

同一スカラ変数が異なる繰り返しで定義・参照されているループ

同一スカラ変数

```
DO I = 1, N
   C(I) = T
   T = B(I)
}
```

- 定義、参照の順であれば並列化可能
- 総和・累積のパターンは、プログラムの変形、指示行の指定などにより並列化可能。(自動並列利用時は、コンパイラが自動的に認識し並列化する)

IF条件下で定義された変数がif条件下以外で参照

```
DO J = 1, N
  DO I = 1, M
    IF (A(I,J) .GE. D ) THEN
      T = A(I,J) - D
    ENDIF

    C(I,J) = T
  ENDDO
ENDDO
```

- 変数**T**は**IF**文の条件が成立した繰り返しで定義された値を参照
- この場合は定義・参照の順でも並列化不可

並列化できない制御構造

ループからの飛び出し

- 飛び出す条件が成立した繰り返しより後の繰り返しを実行してはならないため、並列化できない

```
DO J = 1, N
  DO I = 1, N
    IF (A(I,J) < 0.0 ) GOTO 100
    B(I,J) = SQRT(A(I,J))
  ENDDO
ENDDO
100 CONTINUE
```

指示行による並列化促進

```
$ nfort -mparallel -fdiag-parallel=2 a.f90 -c
nfort: vec( 103): a.f90, line 5: Unvectorized loop.
```

- ループ中に関数呼び出しが含まれるとき、その関数が並列実行可能かどうか不明であるため、ループは自動並列化の対象とならない
- 関数が並列実行可能であれば、**CNCALL**指示行を指定し、ループを自動並列化の対象とする

```
SUBROUTINE SUB(A, B, N)
  INTEGER :: N, I
  REAL :: A(N), B(N), C(N)

  DO I = 1, N
    C(I) = FUNC(A(I), B(I))
  ENDDO
END
```

```
SUBROUTINE SUB(A, B, N)
  INTEGER :: N, I
  REAL :: A(N), B(N), C(N)
!NEC$ CNCALL
  DO I = 1, N
    C(I) = FUNC(A(I), B(I))
  ENDDO
END
```

```
$ nfort -mparallel -fdiag-parallel=2 a.f90 -c
nfort: par(1801): a.f90, line 5: Parallel routine generated.: SUB$1
nfort: par(1803): a.f90, line 5: Parallelized by "do".
nfort: vec( 103): a.f90, line 5: Unvectorized loop.
```

強制並列化指示行

- 自動並列化機能で並列化されなかった
- プログラマはループが並列化可能なことを知っている

- 強制並列化指示行**PARALLEL DO**を指定して並列化
  - ループ、文の集まりを並列化指定できる
  - コンパイラは、データの依存関係を無視して並列化する。

正しい結果が得られることはプログラマが保証しなければならない

```
SUBROUTINE SUB(A, B, N)
  INTEGER :: N, I, J
  REAL :: A(N), B(N), X(N), WK(256)
  REAL :: SUM = 0.0
 !NEC$ PARALLEL DO PRIVATE(WK)
  DO I = 1, N
    DO J = 1, N
      WK(I) = A(I) + B(J)
    ENDDO
    CALL SUB1(X(J), WK)
 !NEC$ ATOMIC
    SUM = SUM + X(J)
  ENDDO
END SUBROUTINE SUB
```

ループの強制並列化を指定
ループ内で作業用として使用する変数、配列は**PRIVATE**句で指定する

強制並列化されたループが、総和、累積などのマクロ演算の文を含むとき、**ATOMIC**をその文の直前に指定する

並列化のためのオーバーヘッド

プログラムを並列化したことにより増加する実行時間のこと

- 並列化するためにプログラマによって追加された処理の実行時間
  - プログラム変形による増加時間
  - 並列処理制御のための実行時ライブラリの処理時間
- システムライブラリ内の排他制御による待ち時間
  - システムデータを更新、参照するシステムライブラリ関数内での排他制御による待ち時間
    - ファイルI/O関数、**MALLOC()**など
- 他のスレッドとのバリア同期のための待ち時間

システムライブラリ内の排他制御

- プログラム全体で利用されるデータを参照、更新するとき、別のOpenMPスレッドで同時に更新されないよう排他制御を行う
  - ファイルディスクリプタ、**MALLOC()**で確保した領域の管理データなど

- システムライブラリ関数の呼び出し回数を削減
  - **MALLOC()**はできるだけ一つにまとめる
  - 関数内でのみ使用するデータは、ローカルデータとして宣言し、スタックに領域を確保する
  - 使用メモリに余裕があるとき、ファイルからの読み込みは一度に行い、内容をメモリに展開し、必要なデータをメモリから読み込むようにする

バリア同期のための待ち時間の削減 (1)

OpenMPでは、次の場所で自動的にバリア同期が行われる

- **NOWAIT**句のない並列ループの終了時
- **REDUCTION**句の指定された並列ループの終了時(*)
- **COPYIN**句の指定された並列リージョンの開始時(*)
- 並列リージョンの終了時(*)

自動並列化では、コンパイラが適切に暗黙的バリア同期を行う

(*)は並列処理の仕組み上、バリア同期を省略することができない

スレッドごとの仕事量を均一にする(待ち時間の短縮)

- 繰り返しごとの作業量が変化する並列ループの作業量の均一化には、**SCHEDULE(DYNAMIC)**が有効

```
!$OMP DO SCHEDULE( STATIC )
DO J = 1, M
  DO I = 1, N
    ...
  ENDDO
ENDDO
```

```
!$OMP DO SCHEDULE( DYNAMIC )
DO J = 1, M
  DO I = 1, N
    ...
  ENDDO
ENDDO
```

バリア同期のための待ち時間の削減 (2)

- 並列リージョンの連結による暗黙的バリア同期の削除
- **NOWAIT**句指定による不要な暗黙的バリア同期の削除
  - 削除できないバリア同期に**NOWAIT**句が指定された場合、コンパイラは**NOWAIT**句の指定を無視する

```
SUBROTINE SUB()
 …
!OMP$ PARALLEL DO
  DO I = 1, N
    A(I) = …
  ENDDO
  X = 0.0
!OMP$ PARALLEL DO
  DO I = 1, N*2
    B(I) = …
  ENDDO
    …
END SUBROTINE SUB
```

並列リージョンの連結

```
SUBROTINE SUB()
 …
  X = 0.0;
!OMP$ PARALLEL
!OMP$ DO
  DO I = 1, N
    A(I) = …
  ENDDO
!OMP$ DO
  DO I = 1, N*2
    B(I) = …
  ENDDO
!OMP$ END PARALLEL
    …
END SUBROTINE SUB
```

NOWAIT句を指定

```
SUBROTINE SUB()
 …
  X = 0.0;
!$OMP PARALLEL
!$OMP DO NOWAIT
  DO I = 1, N
    A(I) = …
  ENDDO
!$OMP DO NOWAIT
  DO I = 1, N*2
    B(I) = …
  ENDDO
!$OMP END PARALLEL
    …
END SUBROTINE SUB
```

ロードバランスの改善 (1)

次のようなループでは、タスクごとの作業量が均一でなく、ループの終了時点で多くの待ち時間が生じる

並列ループを四つに分割し、四つのスレッドで実行したとき




```
!$OMP DO
      DO J = 1024, 1, -1
        DO I = 1, J
           …
        ENDDO
      ENDDO
```

並列化されたループの繰り返しが進むにつれて内側ループの繰り返し数、すなわち、計算量が減少する

スレッドごとの作業量をできるだけ均一にし、待ち時間を少なくすると、より短時間ですべての計算を終えることができる

ロードバランスの改善 (2)

作業量を均一にするため、並列ループをより細かく分割し、スレッドに割り当てる

OpenMP並列化

- **SCHEDULE**句のパラメータで調整

```
!$OMP DO SCHEDULE(DYNAMIC,4)
    DO J = 1024, 1, -1
        DO I = 1, J
            …
        ENDDO
    ENDDO
```

自動並列化

- **CONCURRENT**指示行にOpenMPと同様に**SCHEDULE**句を付けてパラメータで調整

```
!NEC$ CONCURRENT SCHEDULE(DYNAMIC,4)
    DO J = 1024, 1, -1
        DO I = 1, J
            …
        ENDDO
    ENDDO
```



分割数が多いほどスレッド制御のための時間が必要になるので、可能な限り少ない分割数にすること

簡易性能解析機能:FTRACE

スレッドごとの情報より、手続内でのロードバランスを知ることができる

REQUENCY	EXCLUSIVE TIME[sec](%)	AVER.TIME [msec]	MOPS	MFLOPS	V.OP RATIO	AVER. V.LEN	VECTOR TIME	L1CACHE MISS	CPU PORT CONF	VLD LLC HIT	PROC.NAME E.%
60000	62.177(73.1)	1.036	100641.4	79931.0	99.55	248.5	62.134	0.023	0.000	100.00	SUBX$1
15000	4.467(5.3)	0.298	107076.2	83033.3	99.47	248.4	4.455	0.005	0.000	100.00	-thread0
15000	11.552(13.6)	0.770	104082.7	82404.6	99.54	248.5	11.542	0.006	0.000	100.00	-thread1
15000	19.000(22.3)	1.267	101390.4	80683.3	99.55	248.6	18.990	0.006	0.000	100.00	-thread2
15000	27.157(31.9)	1.810	97595.1	77842.2	99.56	248.6	27.147	0.006	0.000	100.00	-thread3
15000	22.711(26.7)	1.514	1426.9	0.0	0.00	0.0	0.000	0.015	0.000	0.00	SUBX
...											
79001	85.034(100.0)	1.076	74062.7	58500.4	98.89	248.5	62.249	0.043	0.000	100.00	total

ループ直前に !NEC$ CONCURRENT SCHEDULE(DYNAMIC, 4) を指定

REQUENCY	EXCLUSIVE TIME[sec](%)	AVER.TIME [msec]	MOPS	MFLOPS	V.OP RATIO	AVER. V.LEN	VECTOR TIME	L1CACHE MISS	CPU PORT CONF	VLD LLC HIT	PROC.NAME E.%
60000	66.872(99.6)	1.115	93599.2	74318.7	99.52	248.5	64.077	1.418	0.000	100.00	SUBX$1
15000	16.766(25.0)	1.118	92992.0	73842.7	99.52	248.5	16.022	0.409	0.000	100.00	-thread0
15000	16.697(24.9)	1.113	91671.0	72790.7	99.52	248.5	16.000	0.397	0.000	100.00	-thread1
15000	16.714(24.9)	1.114	94854.7	75312.8	99.52	248.5	16.040	0.305	0.000	100.00	-thread2
15000	16.695(24.9)	1.113	94880.7	75329.6	99.51	248.5	16.014	0.307	0.000	100.00	-thread3
15000	0.129(0.2)	0.009	1284.5	0.1	0.00	0.0	0.000	0.010	0.000	0.00	SUBX
...											
79001	67.148(100.0)	0.850	93334.5	74082.8	99.51	248.5	64.192	1.430	0.000	100.00	total

改善前：**SUBX$1**の**-thread0**～**-thread3**のEXCLUSIVE TIMEにばらつきがある(ロードインバランス)
改善後：ばらつきがなくなり、**SUBX**のEXCLUSIVE TIMEが短縮(バリア同期時間等が短縮)
　　ただしスレッド制御時間が増すため**SUBX$1**は増加

並列化における注意事項

ALLOCATE文で確保した領域

ALLOCATE文で確保した領域が共有か、プライベートかは以下で決定

- 割付けた配列、またはポインタが共有か、プライベートか?
- 領域確保時、並列処理中だったか?

P,Q : shared
R,S : private

並列処理区間

```
SUBROUTINE SUB()
  REAL,ALLOCATABLE :: P(:), Q(:)
  REAL,ALLOCATABLE :: R(:), S(:)

  ALLOCATE(P(16))
!$OMP PARALLEL PRIVATE(R,S)
!$OMP SINGLE
  ALLOCATE(Q(16))
!$OMP END SINGLE
!$OMP MASTER
  ALLOCATE(R(16))
!$OMP END MASTER
  ALLOCATE(S(16))
!$OMP END PARALLEL
END SUBROUTINE SUB
```

ALLOCATE(P(16))は1回実行。Pはsharedのため、全スレッドで同一の領域を参照する。

ALLOCATE(Q(16))は一つのスレッドでのみ実行され、領域は一つだけ確保される。Qはsharedのため、全スレッドで同一の領域を参照する。

ALLOCATE(R(16))はマスタスレッドでのみ実行され、領域は一つだけ確保される。Rはprivateのため、マスタスレッド以外では未割付けの状態のままである。

ALLOCATE(S(16))は全スレッドで実行され、領域は四つ確保される。Sはprivateのため、各スレッドで別々の領域が使用される。

巨大なローカル配列

- Parallelリージョンなど並列実行される部分において、巨大なローカル配列を使用する場合、環境変数**OMP_STACKSIZE**にその配列より大きいサイズの値を設定してください。
  - **OMP_STACKSIZE**は、マスタスレッド以外のスレッドのスタックの最大サイズを設定する環境変数。設定しなかった場合のスタックの最大サイズは4メガバイト。
  - スタックの未使用領域に配列が入り切らなかったとき、プログラムが異常終了する。




```
$ cat a.f90
PROGRAM MAIN
...
!$OMP PARALLEL
  CALL SUB()
!$OMP END PARALLEL
...
END PROGRAM

SUBROTINE SUB()
  REAL(KIND=8) :: X(16*1024*1024)
  REAL(KIND=8) :: Y(16*1024*1024)
...
END SUBROUTINE SUB
...
$ nfort -fopenmp a.f90
$ export OMP_STACKSIZE=384M
$ ./a.out
```

総和演算

- 総和演算は並列化可能であるが、各スレッドの実行順序が一定でない（実行順序が保証されない）ため、足し込みの順序が実行するたびに変わってしまう可能性がある
  - 演算誤差範囲で、シリアル実行時とは計算結果が異なる、また、並列実行するたびに結果が変わることがある

```
DO I = 1, 100
  SUM = SUM + X(I)
ENDDO
```




Orchestrating a brighter world
NEC